

## Excel. Использование функций СМЕЩ и ДВССЫЛ для поиска

Это седьмая глава книги Билла Джелена. [Всё о ВПР: от первого применения до экспертного уровня.](#)

**Суммирование по диапазону, размер которого вычисляется в ячейке.** Проблема: мне нужно просуммировать значения в диапазоне, начинающемся в ячейке A5, а количество строк диапазона суммирования задано в ячейке C5 (рис. 7.1).

Стратегия: использовать подвижную (гибкую) функцию СМЕЩ, которая позволит вам:

- Начать отсчет из конкретной ячейки или диапазона ячеек;
- Сместить начало отсчета на заданное число строк и столбцов;
- Задать число строк и столбцов диапазона, считая от нового начала отсчета.




Рис. 7.1. Эта формула суммирует значения в диапазоне A5:A9

Первый аргумент функции СМЕЩ задает начало отсчета. В нашем примере – ячейку A5. Второй и третий аргументы позволяют изменить начало отсчета. Это полезно, когда значения этих аргументов берутся из ячеек, в которых выполняются какие-то расчеты. В нашем примере начало отсчета не смещено (мы видим два нуля). Мы могли бы указать в качестве второго аргумента 2, и тогда начало отсчета сместилось бы в A7, или –2, и тогда стартовали бы из ячейки A3. Четвертый и пятый аргументы задают размер диапазона. Они являются необязательными. Если эти аргументы отсутствуют, то высота и ширина возвращаемого диапазона такая же, как и для исходного диапазона в первом аргументе. В нашем примере это пять строк и один столбец.

В общем случае функция СМЕЩ возвращает ссылку на диапазон или, другими словами, массив значений, хранящихся в этом диапазоне (с сохранением формата диапазона, то есть числа строк и столбцов). Если использовать СМЕЩ саму по себе, то в большинстве случаев она вернет некорректный результат (одно из значений массива) или ошибку #ЗНАЧ!. Поэтому СМЕЩ, как правило, используется в качестве аргумента другой функции. В нашем примере – СУММ. Только если СМЕЩ возвращает ссылку на единственную ячейку, то и сама по себе она вернет корректный результат – значение в этой ячейке.

Измените значение в C5 с 5 на 3, и формула вернет сумму значений в диапазоне A5:A7 (рис. 7.2). При этом сама формула в ячейке C7 не поменялась))

`=СУММ(СМЕЩ($A$5;0;0;$C$5;1))`

|    | A  | B | C | D |
|----|----|---|---|---|
| 4  | 29 |   |   |   |
| 5  | 1  |   | 3 |   |
| 6  | 2  |   |   |   |
| 7  | 4  |   | 7 |   |
| 8  | 8  |   |   |   |
| 9  | 16 |   |   |   |
| 10 | 32 |   |   |   |
| 11 | 64 |   |   |   |

Рис. 7.2. Измените значение в C5 на 3, и диапазон суммирования изменится

СМЕЩ может использоваться, чтобы указать на ячейку выше текущей. Зачем вы идете на такие ухищрения, когда простая формула делает то же самое?

B5 : =A4

|   | A       | B                  | C |
|---|---------|--------------------|---|
| 1 | Задание | Предыдущее задание |   |
| 2 | Шаг 1   |                    |   |
| 3 | Шаг 2   | Шаг 1              |   |
| 4 | Шаг 3   | Шаг 2              |   |
| 5 | Шаг 4   | Шаг 3              |   |
| 6 | Шаг 5   | Шаг 4              |   |

Рис. 7.3. Какая потребность в СМЕЩ, если работает простая ссылка =A4

Однако, что произойдет если удалить строку 4? Простая формула в столбце B даст ошибку #ССЫЛКА! Формулы СМЕЩ продолжает работать (рис. 7.4).

E4 : =СМЕЩ(E4;-1;-1)

|   | A       | B                  | C | D       | E                 | F |
|---|---------|--------------------|---|---------|-------------------|---|
| 1 | Задание | Предыдущее задание |   | Задание | Предыдущее задани |   |
| 2 | Шаг 1   |                    |   | Шаг 1   |                   |   |
| 3 | Шаг 2   | Шаг 1              |   | Шаг 2   | Шаг 1             |   |
| 4 | Шаг 4   | #ССЫЛКА!           |   | Шаг 4   | Шаг 2             |   |
| 5 | Шаг 5   | Шаг 4              |   | Шаг 5   | Шаг 4             |   |

Рис. 7.4. Если мы предполагаем, что какая-то строка может быть удалена, то функция СМЕЩ незаменима

Дополнительные сведения: начальный диапазон может быть больше чем одна ячейка. В следующем примере начальный диапазон A4:A13. Третий аргумент функции СМЕЩ является вычисляемым – МЕСЯЦ(A1). Благодаря ему диапазон суммирования смещается на 5 столбцов вправо. Формула =СУММ(СМЕЩ(A4:A13;0;МЕСЯЦ(A1))) суммирует столбец, соответствующий дате в ячейке A1.

A2 : =СУММ(СМЕЩ(A4:A13;0;МЕСЯЦ(A1)))

|    | A               | B   | C   | D   | E   | F   | G   | H   |
|----|-----------------|-----|-----|-----|-----|-----|-----|-----|
| 1  | 06.05.2015      |     |     |     |     |     |     |     |
| 2  | 568             |     |     |     |     |     |     |     |
| 3  |                 | Янв | Фев | Мар | Апр | Май | Июн | Июл |
| 4  | Москва          | 73  | 63  | 30  | 82  | 90  | 87  | 77  |
| 5  | Санкт-Петербург | 88  | 77  | 81  | 72  | 49  | 66  | 31  |
| 6  | Новосибирск     | 23  | 65  | 80  | 69  | 80  | 44  | 32  |
| 7  | Екатеринбург    | 79  | 65  | 49  | 47  | 90  | 74  | 51  |
| 8  | Н.-Новгород     | 23  | 45  | 54  | 72  | 29  | 62  | 25  |
| 9  | Казань          | 24  | 80  | 89  | 88  | 36  | 86  | 79  |
| 10 | Челябинск       | 49  | 73  | 29  | 88  | 67  | 73  | 31  |
| 11 | Омск            | 89  | 24  | 85  | 45  | 35  | 84  | 83  |
| 12 | Самара          | 24  | 30  | 90  | 77  | 65  | 67  | 66  |
| 13 | Ростов-на-Дону  | 79  | 58  | 65  | 86  | 27  | 67  | 36  |
| 14 |                 |     |     |     |     |     |     |     |
| 15 |                 |     |     |     |     |     |     |     |
| 16 |                 |     |     |     |     |     |     |     |

Рис. 7.5. Используйте СМЕЩ для перемещения диапазона суммирования на N столбцов вправо

К сожалению, функция СМЕЩ является волатильной (переменной) функцией Excel. Это означает, что с каждым вычислением рабочего листа, СМЕЩ пересчитывается, даже если ни одна ячейка, связанная с ней, не изменилась. Если у вас много функций СМЕЩ, это может вызвать замедление работы Excel.

Можно использовать ИНДЕКС вместо СМЕЩ (рис. 7.6). В принципе, синтаксис функции =ИНДЕКС(массив; номер\_строки; [номер\_столбца]) предназначен для возврата значения одной ячейки из диапазона. Однако, если вы опустите второй аргумент, Excel вернет все строки, т.е. ссылку на диапазон F4:F13 (или массив {90:49:80:90:29:36:67:35:65:27}) Формула =СУММ(ИНДЕКС(B4:H13;;МЕСЯЦ(A1))) вернет результат эквивалентный формуле на рис. 7.5.

A2 : =СУММ(ИНДЕКС(B4:H13;;МЕСЯЦ(A1)))

|    | A               | B   | C   | D   | E   | F   | G   | H   |
|----|-----------------|-----|-----|-----|-----|-----|-----|-----|
| 1  | 06.05.2015      |     |     |     |     |     |     |     |
| 2  | 568             |     |     |     |     |     |     |     |
| 3  |                 | Янв | Фев | Мар | Апр | Май | Июн | Июл |
| 4  | Москва          | 73  | 63  | 30  | 82  | 90  | 87  | 77  |
| 5  | Санкт-Петербург | 88  | 77  | 81  | 72  | 49  | 66  | 31  |
| 6  | Новосибирск     | 23  | 65  | 80  | 69  | 80  | 44  | 32  |
| 7  | Екатеринбург    | 79  | 65  | 49  | 47  | 90  | 74  | 51  |
| 8  | Н.-Новгород     | 23  | 45  | 54  | 72  | 29  | 62  | 25  |
| 9  | Казань          | 24  | 80  | 89  | 88  | 36  | 86  | 79  |
| 10 | Челябинск       | 49  | 73  | 29  | 88  | 67  | 73  | 31  |
| 11 | Омск            | 89  | 24  | 85  | 45  | 35  | 84  | 83  |
| 12 | Самара          | 24  | 30  | 90  | 77  | 65  | 67  | 66  |
| 13 | Ростов-на-Дону  | 79  | 58  | 65  | 86  | 27  | 67  | 36  |
| 14 |                 |     |     |     |     |     |     |     |
| 15 |                 |     |     |     |     |     |     |     |
| 16 |                 |     |     |     |     |     |     |     |

Рис. 7.6. ИНДЕКС с опущенным аргументом Номер\_строки вернет все строки диапазона, т.е. весь столбец

**Что случилось с функцией @?@?** В Lotus 1-2-3 была функция @@. Если вы использовали @@(A3), Lotus пошел бы в A3. A3 должна была содержать ссылку на ячейку. Допустим, в A3 содержался текст C5. Функция @@ возвращала значение из ячейки C5.

Стратегия: в Excel также есть аналогичная функция – ДВССЫЛ. Вот несколько примеров как она работает. Рассмотрим формулу =ДВССЫЛ(F2). Excel пойдет в F2 и использует адрес ячейки, найденный там (рис. 7.7). Формула сначала смотрит в F4, а потом в C1. Имена ячеек должны быть набраны английскими буквами, иначе ДВССЫЛ вернет ошибку #ССЫЛКА! Ссылку на ячейку в функции ДВССЫЛ рассчитывается «на лету».

F4 : =ДВССЫЛ(F2)

|   | A  | B  | C  | D  | E | F  |
|---|----|----|----|----|---|----|
| 1 | 18 | 19 | 3  | 13 |   |    |
| 2 | 4  | 15 | 4  | 21 |   | C1 |
| 3 | 30 | 4  | 22 | 12 |   |    |
| 4 | 17 | 9  | 12 | 23 |   | 3  |
| 5 | 9  | 13 | 16 | 1  |   |    |
| 6 | 5  | 23 | 12 | 28 |   |    |
| 7 | 21 | 4  | 14 | 11 |   |    |
| 8 | 4  | 8  | 6  | 25 |   |    |
| 9 |    |    |    |    |   |    |

Рис. 7.7. F2 говорит: посмотри в ячейку C1

В следующем примере (рис. 7.8) функция ВПР указывает на другой лист, основанный на номере квартала в столбце B. ДВССЫЛ использует конкатенацию (&), применяемую для сцепления текстовых строк, чтобы «на лету» построить имя листа в книге Excel.

=ВПР(A2;ДВССЫЛ("Q"&B2&"!A1:B99");2;ЛОЖЬ)

|   | A     | B       | C    | D | E |
|---|-------|---------|------|---|---|
| 1 | Центр | Квартал | Итог |   |   |
| 2 | 1060  | 2       | 200  |   |   |
| 3 | 1060  | 1       | 100  |   |   |
| 4 | 1060  | 3       | 300  |   |   |
| 5 | 1030  | 4       | 400  |   |   |
| 6 | 1040  | 4       | 400  |   |   |
| 7 | 1060  | ?       | 200  |   |   |

Sheet1 Report Q1 Q2 Q3 Q4

Рис. 7.8. Имя листа Q2! рассчитывается «на лету» внутри функции ДВССЫЛ

Если вы используете имена диапазонов, значение внутри ДВССЫЛ может указывать на имя диапазона. Это создает некоторые интересные возможности поиска. Сначала назначьте имена диапазонам. Для этого выберите диапазон целиком (рис. 7.9). На вкладке ФОРМУЛЫ выберите команду создать из выделенного (или нажмите Ctrl+Shift+F3).

ФОРМУЛЫ

Диспетчер имен

Присвоить имя

Использовать в формуле

Создать из выделенного

Определенные имена

A1 : fx

|   | A     | B   | C   | D   | E   | F   | G   | H  | I   |
|---|-------|-----|-----|-----|-----|-----|-----|----|-----|
| 1 |       | Янв | Фев | Мар | Апр | Май | Июн | Юл | Авг |
| 2 | Прод1 | 1   | 2   | 3   | 5   | 8   | 13  | 21 | 34  |
| 3 | Прод2 | 2   | 3   | 5   | 8   | 13  | 21  | 34 | 55  |
| 4 | Прод3 | 3   | 4   | 7   | 11  | 18  | 29  | 47 | 76  |
| 5 | Прод4 | 4   | 5   | 9   | 14  | 23  | 37  | 60 | 97  |
| 6 | Прод5 | 5   | 6   | 11  | 17  | 28  | 45  | 73 | 118 |
| 7 | Прод6 | 6   | 7   | 13  | 20  | 33  | 53  | 86 | 139 |
| 8 | Прод7 | 7   | 8   | 15  | 23  | 38  | 61  | 99 | 160 |

Рис. 7.9. Выберите диапазон подстановки, включая заголовки

В открывшемся окне *Создание имен из выделенного диапазона* укажите, что имена диапазонов нужно брать из заголовков таблицы, расположенных в верхней строке и левом столбце.




Рис. 7.10. Установка параметров в окне *Создание имен из выделенного диапазона*

Автоматически сгенерируются имена для строк и столбцов. Например, диапазон \$B\$3:\$I\$3 получит имя *Прод2*, а диапазон \$E\$2:\$E\$8 имя *Апр*. Все имена диапазонов можно увидеть, пройдя по меню *ФОРМУЛЫ* → *Диспетчер имен* (рис. 7.11).




Рис. 7.11. Автоматически созданные имен диапазонов в окне *Диспетчера имен*

Теперь у вас есть семь именованных диапазонов на основе меток (имен) в столбце А и восемь именованные диапазоны, основанные на заголовках столбцов.

И наконец, вы можете использовать функцию СУММ, чтобы найти значение на пересечении заданных строки и столбца. Как? Поясню. Очень многие пользователи применяют функцию СУММ, в которой аргументы разделены точкой с запятой или двоеточием. Например, СУММ(A2;B4) подсчитает сумму в ячейках A2 и B4, а СУММ(A2:B4) подсчитает сумму в диапазоне ячеек A2:B4 (всего 6 ячеек). Но есть еще одна мало известная возможность – разделить ссылки на диапазоны пробелом (рис. 7.12). Формула =СУММ(ДВССЫЛ(J10) ДВССЫЛ(J11)) сначала «на лету» вычислит значения ДВССЫЛ(J10) = Прод4 и ДВССЫЛ(J11) = Апр. При этом значения *Прод4* и *Апр* функция СУММ будет воспринимать, как имена диапазонов. Поскольку использован оператор пробела, функция СУММ вернет значение на пересечении диапазонов *Прод4* и *Апр*, т.е. значение из ячейки E5.

|    | A     | B   | C   | D   | E   | F   | G   | H   | I   | J              | K | L |
|----|-------|-----|-----|-----|-----|-----|-----|-----|-----|----------------|---|---|
| 1  |       | Янв | Фев | Мар | Апр | Май | Июн | Июл | Авг |                |   |   |
| 2  | Прод1 | 1   | 2   | 3   | 5   | 8   | 13  | 21  | 34  |                |   |   |
| 3  | Прод2 | 2   | 3   | 5   | 8   | 13  | 21  | 34  | 55  |                |   |   |
| 4  | Прод3 | 3   | 4   | 7   | 11  | 18  | 29  | 47  | 76  |                |   |   |
| 5  | Прод4 | 4   | 5   | 9   | 14  | 23  | 37  | 60  | 97  |                |   |   |
| 6  | Прод5 | 5   | 6   | 11  | 17  | 28  | 45  | 73  | 118 |                |   |   |
| 7  | Прод6 | 6   | 7   | 13  | 20  | 33  | 53  | 86  | 139 |                |   |   |
| 8  | Прод7 | 7   | 8   | 15  | 23  | 38  | 61  | 99  | 160 |                |   |   |
| 9  |       |     |     |     |     |     |     |     |     |                |   |   |
| 10 |       |     |     |     |     |     |     |     |     | Артикул: Прод4 |   |   |
| 11 |       |     |     |     |     |     |     |     |     | Месяц: Апр     |   |   |
| 12 |       |     |     |     |     |     |     |     |     | Итого: 14      |   |   |
| 13 |       |     |     |     |     |     |     |     |     |                |   |   |
| 14 |       |     |     |     |     |     |     |     |     |                |   |   |
| 15 |       |     |     |     |     |     |     |     |     |                |   |   |

Рис. 7.12. Необычный способ поиска по двум направлениям

См. также [Примеры использования функции ДВССЫЛ \(INDIRECT\)](#), [Примеры использования функции СМЕЩ \(OFFSET\)](#), [Динамические диапазоны на основе функций ИНДЕКС и СМЕЩ](#).